
Introduction

Zygo Corporation provides an ActiveX DLL to allow programmers to easily read and write MetroPro data files.  This DLL can be used by any ActiveX enable programming language.  This document describes the interfaces exposed by the DLL, and provides example code.

Overview

This DLL provides the facilities to perform the following operations:

· Read a standard MetroPro data file containing phase and/or intensity data.

· Get information about the data file from the header and attributes.

· Set information about the data file.

· Save a standard MetroPro data file containing phase and/or intensity data.

· Use of ActiveX objects to manipulate data sets.

· Backward compatibility with the MPVB DLL.

The objects exposed by the DLL each support the manipulation of a single MetroPro data set at a time.  However, multiple objects can be created to manipulate multiple data sets simultaneously.

Data Types

The following data types are used in this document.  The Visual Basic equivalent data types follow each generic description.  This document uses the Visual Basic names for the data types.

	Generic Description
	Visual Basic
	C/C++

	2‑byte signed integer
	Integer
	short

	4‑byte signed integer
	Long
	long

	4‑byte floating point
	Single
	float

	Length prefixed wide character string
	String
	BSTR

	Boolean
	Boolean
	VARIANT_BOOL


Contents of the DLL

This section describes the enumerated types and objects exposed by the DLL.

Enumerated Types

Enumerated types is a user‑defined type consisting of named constants.  They are used to make the code more readable when each integer value has a specific meaning.

ZygoSoftwareType

Used by the Type property of ZygoSoftwareInfo.

	swtUnknown
	0

	swtMetroPro
	1

	swtMetroBasic
	2

	swtDiagnostic
	3


ZygoSourceAttr

Used by the Source property of ZygoDataAttributes.

	UnknownSource
	-1

	Measured
	0

	Generated
	1

	Scanned
	2


ZygoAcquireType

Used by the AcquireType property of ZygoDataAttributes.

	AcqPhase
	0

	AcqFringe
	1

	AcqScan
	2

	AcqIntensity
	3

	AcqBright
	4

	AcqbrightDark
	5

	AcqFromFile
	6


ZygoResolution

Used by the PhaseRes property of ZygoDataAttributes.

	UnknownRes
	-1

	LowRes
	0

	HighRes
	1

	SuperRes
	2


ZygoDisconAction

Used by the DisconAction property of ZygoDataAttributes.

	DeleteRegionsWithDiscontinuities
	0

	DeleteDiscontinuities
	1

	IgnoreDiscontinuities
	2

	DeleteWithoutChecking
	3

	SuperDeleteDiscontinuities
	4


ZygoConnectionOrder

Used by the ConnectionOrder property of ZygoDataAttributes.

	ConnectByLocation
	0

	ConnectBySize
	1


ZygoSystemType

Used by the SystemType property of ZygoDataAttributes.

	sysUnknown
	0

	sysMark
	1

	sysMaxim
	2

	sysMaximNT
	3

	sysGPI
	4

	sysMaximXR
	5

	sysMaximGP
	6

	sysMaximXRGP
	7

	sysMarkGPI
	8

	sysRVA
	9

	sysMaximVS
	10

	sysMaximXRICO
	11

	sysMaximGPICO
	12

	sysMaximXRGPICO
	13

	sysMesa
	14

	sysNIF
	15

	sysFTP
	16

	sysMicroLupi
	17

	sysDVD
	18

	sysFTPSI
	19

	sysMaximDelta
	20

	sysSAI
	21

	sysNewview6000
	22

	sysGpiCan
	23

	sysMarkFP
	24


ZygoCodeVType

Used by the CodeVType property of ZygoDataAttributes.

	codeVWavefront
	0

	codeVSurface
	1


ZygoFdaRes

Used by the FDARes property of ZygoDataAttributes.

	fdaResTest1
	-100

	fdaResTest2
	-101

	fdaResSmooth
	-4

	fdaResCentroid
	-3

	fdaResNone
	-2

	fdaResFineOld
	-1

	fdaResMedium
	0

	fdaResFine
	1

	fdaResHigh2g
	3

	fdaResDissMin
	8

	fdaResDissMax
	9

	fdaResPhase1
	20

	fdaResPhase2
	21

	fdaResPhase3
	22


ZygoCoherenceMode

Used by the CoherenceMode property of ZygoDataAttributes.

	cohModeNone
	0

	cohModeStandard
	1

	cohModeShort
	2


ZygoSurfaceFilter

Used by the CoherenceMode property of ZygoDataAttributes.

	surfFilterNone
	0

	surfFilterStandard
	1

	surfFilterRough
	2


ZygoOrientation

Used by the RcOrientation property of ZygoDataAttributes2.

	zoUnknown
	32 (‘ ‘)

	zoNorth
	78 (‘N’)

	zoSouth
	83 (‘S’)

	zoEast
	69 (‘E’)

	zoWest
	87 (‘W’)


ZygoFtpsiPhaseRes

Used by the FtpsiPhaseResolution property of ZygoDataAttributes2.

	zfprHigh
	0

	zfprMax
	1

	zfprCustom
	2


ZygoCavityType

Used by the CavityType property of ZygoDataAttributes2.

	zctThin
	0

	zctThick
	1

	zctMedium
	2

	zctAuto
	3


ZygoScanDirection

Used by the ScanDirection property of ZygoDataAttributes2.

	zsdNode
	0

	zsdUpward
	1

	zsdDownward
	2


Objects

ZygoSoftwareVersion (IZygoSoftwareVersion interface)
Properties

	Major as Integer
	Major version number.

	Minor as Integer
	Minor version number.

	Bug as Integer
	Bug version number.


Methods

	Copy(Source as ZygoSoftwareVersion)
	Copies the values of Source.


ZygoSoftwareInfo (IZygoSoftwareInfo interface)
Properties

	Type as ZygoSoftwareType
	Indicates which program created the data file.

	Date as String
	The date the data file was created in the format “Tue Apr 23 14:13:48 EDT 2002”.  The value is truncated to 29 characters when the data file is saved.

	Version as ZygoSoftwareVersion
	The version number of the program that created the data file.


Methods

	Copy(Source as ZygoSoftwareInfo)
	Copies the values of Source.


ZygoDataLocation (IZygoDataLocation interface)
Properties

	XPosition as Single
	The position of the X‑axis when the data was measured in inches.

	YPosition as Single
	The position of the Y‑axis when the data was measured in inches.

	ZPosition as Single
	The position of the Z‑axis when the data was measured in inches.

	XRotation as Single
	The rotation of the X‑axis (Roll) when the data was measured in degrees.

	YRotation as Single
	The rotation of the Y‑axis (Pitch) when the data was measured in degrees.

	ZRotation as Single
	The rotation of the Z‑axis (Theta) when the data was measured in degrees.


Methods

	Copy(Source as ZygoDataLocation)
	Copies the values of Source.


ZygoDataAttributes (IZygoDataAttributes interface)
Properties

	TimeStamp as Long
	The date and time the data was measured.

	Comment as String
	User‑entered remark line.  The value is truncated to 81 characters when the data file is saved.

	Source as ZygoSourceAttr
	Indicates how the data was acquired or generated.

	IntfScaleFactor as Single
	The interferometric scale factor is the number of waves per fringe.

	WavelengthIn as Single
	The wavelength, in meters, at which the interferogram was measured.

	ObliquityFactor as Single
	Phase correction factor required when using a Mirau objective on a microscope.  A value of 1.0 indicates no correction.

	NumericAperture as Single
	Equals 1 / (2 * f‑number).

	Magnification as Single
	Reserved for future use.

	LateralResolution as Single
	Lateral resolving power of a camera pixel in meters/pixel.  A value of 0 means that the value is unknown.

	AcquireType as ZygoAcquireType
	Indicates the setting of the Acquisition Mode control.

	IntensityAverageCount as Integer
	The number of intensity averages performed.  Values of 0 or 1 indicate no averaging.

	PZTCal as Boolean
	Indicates whether or not the modulation amplitude was automatically adjusted during acquisition.

	PZTGainTolerance as Integer
	Specifies the PZT error range if PZT calibration was adjusted.

	PZTGain as Integer
	Specifies the modulation amplitude value used during data acquisition.

	PartThickness as Single
	The thickness, in meters, of the part measured.

	AGC as Boolean
	Indicates whether or not automatic gain control was performed during data acquisition.

	TargetRange as Single
	The acceptable tolerance limits of the light intensity used during AGC.

	RadCrvMeasSeq as Integer
	

	MinMod as Long
	Minimum value of modulation needed to calculate a phase value.

	MinModCount as Long
	Minimum number of data points required to pass MinMod criteria during AGC.

	PhaseRes as ZygoResolution.
	The resolution of the phase data points.  LowRes means that each fringe is represented by 4096 counts.  HighRes means that each fringe is represented by 32768 counts.

	MinAreaSize as Long
	Minimum number of contiguous data points required for a valid region.

	DisconAction as ZygoDisconAction
	Indicates the action taken when the system encounters discontinuities in phase data.

	DisconFilter as Single
	Specifies the degree to which discontinuities were removed when DisconAction was filtered.

	ConnectionOrder as ZygoConnectionOrder
	Order in which separate regions of phase data are processed.

	DataSign as Integer
	Sign of the data.  0 is normal, 1 is inverted.

	CameraWidth as Integer
	The width (columns) of the usable camera field in pixels.

	CameraHeight as Integer
	The height (rows) of the usable camera field in pixels.

	SystemType as ZygoSystemType
	Indicates the system used to make the measurement.

	SystemBoard as Integer
	Indicates which system board was used when the data measurement was taken.  Valid values range from 0 to 7.

	SystemSerial as Integer
	The serial number of the instrument.

	InstrumentID as Integer
	The instrument unit number.  Valid values range from 0 to 7.

	ObjectiveName as String
	For microscopes, this is the objective in use.  For the GPI, this is the aperture in use.  For generated data, this is blank.  The value is truncated to 11 characters when the data file is saved.

	PartName as String
	User‑entered identified for the part measured.  The value is truncated to 39 characters when the data file is saved.

	CodeVType as ZygoCodeVType
	Indicates whether the phase data represents a wavefront or a surface.  This is only used by the CODE V program.

	PhaseAverageCount as Integer
	The number of phase averages performed.

	SubtractSystemError as Boolean
	Indicates whether or not the system error was subtracted from the phase data.

	PartSerialNumber as String
	User‑entered serial number for the part.  The value is truncated to 39 characters when the data file is saved.

	RefractiveIndex as Single
	Index of refraction as specified by the user.  This is used only in the calculation of corner cube dihedral angles.

	RemoveTiltBias as Boolean
	Indicates whether or not the tilt bias was removed from the phase data.

	RemoveFringes s Boolean
	Indicates whether or not fringes were removed from the intensity data.

	MaxAreaSize as Long
	Maximum number of contiguous data points required for a valid phase data region.  Any larger regions are deleted.

	SetupType as Integer
	Identifies the different setup used by the Angle Measurement Application.

	Wrapped as Boolean
	

	PreConnectFilter as Single
	The amount of filtering done to wrapped phase data.

	WavelengthIn2 as Single
	The value from the Wavelength‑In 2 control, if relevant to the measurement.

	WavelengthFold as Boolean
	Indicates whether or not two wavelength folding is on.

	WavelengthIn1 as Single
	The value from the Wavelength‑In 1 control, if relevant to the measurement.

	WavelengthIn3 as Single
	The value from the Wavelength‑In 3 control, if relevant to the measurement.

	WavelengthIn4 as Single
	The value from the Wavelength‑In 4 control, if relevant to the measurement.

	WavelengthSelect as String
	The value of the Wavelength Select control, if relevant to the measurement.  The value is truncated to 7 characters when the data file is saved.

	FDARes as ZygoFdaRes
	Indicates the setting of the FDA Res control for a measurement when the Acquisition Mode control is set to Scan.

	ScanDescription as String
	Description of the scan for a measurement when the Acquisition Mode control is set to Scan.  The value is truncated to 19 characters when the data file is saved.  A sample string is “5 um bipolar”.

	NFiducials as Integer
	The number of fiducials defined.

	FiducialX(Index as Long) as Single
	The X component of the fiducial’s coordinates specified by Index.  Index can range from 1 to 7.  The coordinate system is the camera’s and (0,0) is in the upper left corner of the video monitor.

	FiducialY(Index as Long) as Single
	The Y component of the fiducial’s coordinates specified by Index.  Index can range from 1 to 7.  The coordinate system is the camera’s and (0,0) is in the upper left corner of the video monitor.

	PixelWidth as Single
	The effective camera pixel width (spacing) in meters.  Effects the scaling of Mask Editor figures.  If zero, default values are assumed.  Not the same as LateralResolution.

	PixelHeight as Single
	The effective camera pixel height  (spacing) in meters.  Effects the scaling of Mask Editor figures.  If zero, default values are assumed.  Not the same as LateralResolution.

	ExitPupilDiameter as Single
	The exit pupil diameter setting used during data acquisition.

	LightLevelPercent as Single
	The light level percent used during data acquisition.

	CoordinateState as Long
	Indicates whether or not the coordinates in the Coordinates property are valid.  0 indicates invalid coordinates.

	Coordinates as ZygoDataLocation
	Coordinates of the stage during data acquisition.

	CoherenceMode as ZygoCoherenceMode
	The coherence mode setting used during data acquisition.

	SurfaceFilter as ZygoSurfaceFilter
	The surface filter setting used during data acquisition.

	SystemErrorFileName as String
	User‑entered name of the file containing the system error data.  The value is truncated to 27 characters when the data file is saved.

	ZoomDescription as String
	Value of the image zoom used during data acquisition.  The value is truncated to 7 characters when the data file is saved.

	AplhaPart as Single
	

	BetaPart as Single
	

	DistancePart as Single
	

	CameraSplitLocationX as Integer
	

	CameraSplitLocationY as Integer
	

	CameraSplitTransX as Integer
	

	CameraSplitTransY as Integer
	

	MaterialA as String
	The value is truncated to 23 characters when the data file is saved.

	MaterialB as String
	The value is truncated to 23 characters when the data file is saved.

	CameraSplitUnused as Integer
	

	DMICenterX as Single
	

	DMICenterY as Single
	

	SphDistCorr as Boolean
	

	SphDistPartNa as Single
	

	SphDistPartRadius as Single
	

	SphDistCalNa as Single
	

	SphDistCalRadius as Single
	

	SurfaceType as Integer
	

	ZPosition as Single
	

	PowerMultiplier as Single
	

	FocusMultiplier as Single
	

	RadCrvFocusCalFactor as Single
	

	RadCrvPowerCalFactor as Single
	


Methods

	Copy(Source as ZygoDataAttributes)
	Copies the values of Source.


ZygoDataAttributes  (IZygoDataAttributes2 interface)
Properties

	AcSurfaceType as Integer
	

	CalPixelLocationX as Integer
	X coordinate of cal pixel in camera coordinates

	CalPixelLocationY as Integer
	Y coordinate of cal pixel in camera coordinates

	CameraFrameRate as Single
	

	CavityType as ZygoZavityType
	

	FilmsDataFormat as Integer
	

	FilmsMaxReflRatio as Single
	

	FilmsMaxThickness as Single
	

	FilmsMinMod as Single
	

	FilmsMinReflRatio as Single
	

	FilmsMinThickness as Single
	

	FilmsMode as Integer
	

	FilmsObliquityCorrection as Single
	

	FilmsReflectivityRatio as Integer
	

	FilmsRefractionIndex as Single
	

	FilmsSystemCharacterizationFileName as String
	

	FilmsMeritMode as Integer
	

	FilmsH2G as Integer
	

	FlashPhaseAliasMask as Single
	

	FlashPhaseDcMask as Single
	

	FlashPhaseFilter as Single
	

	FramesAcquired as Integer
	

	FtpLeftPosition as Single
	

	FtpPitchPosition as Single
	

	FtpRightPosition as Single
	

	FtpRollPosition as Single
	

	FtpsiPhaseResolution as ZygoFtpsiPhaseRes
	

	FtpsiResolutionFactor as Single
	

	MaxIntensity as Long
	Maximum intensity from frame grabber

	MinModPercent as Single
	Min Mod as a percent

	RcAngle as Single
	

	RcDiameter as Single
	

	RcDistance as Single
	

	RcOrientation as ZygoOrientation
	

	ReferenceCalPixelOpd as Single
	

	ReferenceCalPointCount as Integer
	

	ReferenceCalPointX as Single
	

	ReferenceCalPointY as Single
	

	RemoveFringesMode as Boolean
	

	RingOfFire as Boolean
	

	ScanDirection as ZygoScanDirection
	

	SystemSerial2 as Long
	

	TestCalPixelOpd as Single
	

	TestCalPointCount as Integer
	

	TestCalPointX as Single
	

	TestCalPointY as Single
	

	TuneRange as Single
	Optical frequency excursion (Hz)


Methods

	Copy(Source as IZygoDataAttributes2)
	Copies the values of Source.


ZygoDataFileHeader (IZygoDataFileHeader interface)
Properties

	MagicNumber as Long
	Indicates whether or not the file is a valid MetroPro file.  The file is valid only if the value is –2011495569 (881B036F hex).

	HeaderFormat as Integer
	Indicates the format of the header.  This is always 1.

	HeaderSize as Long
	The size of the header in bytes.  This should always be 834.

	SoftwareInfo as ZygoSoftwareInfo
	Software date, type and version.

	IntensOriginX as Integer
	The X component of the origin of the intensity data matrix.  The coordinate system is the camera’s and (0,0) is in the upper left corner of the video monitor.

	IntensOriginY as Integer
	The Y component of the origin of the intensity data matrix.  The coordinate system is the camera’s and (0,0) is in the upper left corner of the video monitor.

	IntensWidth as Integer
	The width (number of columns) of the intensity data matrix.  If no intensity data matrix is present, then this value is zero.

	IntensHeight as Integer
	The height (number of rows) of the intensity data matrix.  If no intensity data matrix is present, then this value is zero.

	IntensNumBuckets as Integer
	The number of buckets of intensity data that are stored.  If no intensity data matrix is present, then this value is zero.

	IntensRange as Integer
	Maximum possible value of an intensity data point.

	IntensNumBytes as Long
	Number of bytes of intensity data stored.  This value is equal to IntensWidth * IntensHeight * IntensNumBuckets * 2.

	PhaseOriginX as Integer
	The X component of the origin of the phase data matrix.  The coordinate system is the camera’s and (0,0) is in the upper left corner of the video monitor.

	PhaseOriginY as Integer
	The Y component of the origin of the phase data matrix.  The coordinate system is the camera’s and (0,0) is in the upper left corner of the video monitor.

	PhaseWidth as Integer
	The width (number of columns) of the phase data matrix.  If no phase data matrix is present, then this value is zero.

	PhaseHeight as Integer
	The height (number of rows) of the phase data matrix.  If no phase data matrix is present, then this value is zero.

	PhaseNumBytes as Long
	Number of bytes of phase data stored.  This value is equal to PhaseWidth * PhaseHeight * 4.

	Attr as ZygoDataAttributes
	Attributes of the data.


Methods

	Copy(Source as ZygoDataFileHeader)
	Copies the values of Source.


ZygoDataWindow (IZygoDataWindow interface)
Properties

	X as Integer
	The X component of the origin of the data matrix.

	Y as Integer
	The Y component of the origin of the data matrix.

	Width as Integer
	The width (number of columns) of the data matrix.

	Height as Integer
	The height (number of rows) of the data matrix.


Methods

	Copy(Source as ZygoDataWindow)
	Copies the values of Source.


ZygoDataMatrix (IZygoDataMatrix interface)
Properties

	Window as ZygoDataWindow
	Specifies the origin and size of the data matrix.  The X, Y, Width, and Height values are the same as PhaseOriginX, PhaseOriginY, PhaseWidth, and PhaseHeight in ZygoDataFileHeader.

	Matrix as Variant
	2D array of Longs representing the phase data.  The first dimension is the columns, and the second is the rows.  Matrix(column, row).


Methods

	Copy(Source as ZygoDataMatrix)
	Copies the values of Source.


ZygoIntensityMatrix (IZygoIntensityMatrix interface)
Properties

	Window as ZygoDataWindow
	Specifies the origin and size of the data matrix.  The X, Y, Width, and Height values are the same as IntensOriginX, IntensOriginY, IntensWidth, and IntensHeight in ZygoDataFileHeader.

	Matrix as Variant
	3D array of Integers representing the intensity data.  The first dimension is the columns, the second is the rows and the third is the buckets.  Matrix(column, row, bucket).

	Range as Integer
	Maximum possible value of an intensity data point.  This value is the same as IntensRange in ZygoDataFileHeader.

	NumBuckets as Integer
	Read Only.  The number of buckets of intensity data that are stored.  This value is the same as IntensNumBuckets in ZygoDataFileHeader.


Methods

	Copy(Source as ZygoIntensityMatrix)
	Copies the values of Source.


ZygoDataFile (IZygoDataFile interface)
Properties

	Header as ZygoDataFileHeader
	Data file header.

	Attribute as ZygoDataAttributes
	Data file attributes.

	PhaseData as ZygoDataMatrix
	Phase data.

	IntensityData as ZygoIntensityMatrix
	Intensity data.

	InvalidPhase as Long
	Read Only.  Minimum value of phase data that indicates the data point is not valid.


Methods

	LoadData(FileName as String)
	Loads the data file specified by FileName.

	SaveData(FileName as String)
	Saves the current information to the data file specified by FileName.  When the data is saved, the values for the phase and intensity origin and dimensions in ZygoDataFileHeader are overwritten with the values from ZygoDataMatrix and ZygoIntensityMatrix.

	Clear()
	Removes phase and intensity data and resets all header and attribute values to their default values.


Objects for Backward Compatibility

These objects are provided for the purpose of backward compatibility with the MPVB DLL.  In the MPVB DLL, MetroProDataAttr was defined in a Visual Basic Module as a public structure using the Type keyword.  That module also defined the calls into the MPVB DLL using the Declare keyword.  The following objects provide the same functionality while requiring few code changes.

MetroProDataAttr (IMetroProDataAttr interface)
Properties

	CameraCols as Integer
	The number of columns (width) of the usable camera field in pixels.

	CameraRows as Integer
	The number of rows (height) of the usable camera field in pixels.

	CameraRes as Single
	Lateral resolving power of a camera pixel in meters/pixel.  A value of 0 means that the value is unknown.

	IntensMatOrgX as Integer
	The X component of the origin of the intensity data matrix.  The coordinate system is the camera’s and (0,0) is in the upper left corner of the video monitor.

	IntensMatOrgY as Integer
	The Y component of the origin of the intensity data matrix.  The coordinate system is the camera’s and (0,0) is in the upper left corner of the video monitor.

	IntensMatCols as Integer
	The number of columns (width) of the intensity data matrix.  If no intensity data matrix is present, then this value is zero.

	IntensMatRows as Integer
	The number of rows (height) of the intensity data matrix.  If no intensity data matrix is present, then this value is zero.

	IntensMatBuckets as Integer
	The number of buckets of intensity data that are stored.  If no intensity data matrix is present, then this value is zero.

	PhaseMatOrgX as Integer
	The X component of the origin of the phase data matrix.  The coordinate system is the camera’s and (0,0) is in the upper left corner of the video monitor.

	PhaseMatOrgY as Integer
	The Y component of the origin of the phase data matrix.  The coordinate system is the camera’s and (0,0) is in the upper left corner of the video monitor.

	PhaseMatCols as Integer
	The number of columns (width) of the phase data matrix.  If no phase data matrix is present, then this value is zero.

	PhaseMatRows as Integer
	The number of rows (height) of the phase data matrix.  If no phase data matrix is present, then this value is zero.

	PhaseRes as Integer
	The resolution of the phase data points.  0 means that each fringe is represented by 4096 counts.  1 means that each fringe is represented by 32768 counts.

	IntfScaleFactor as Single
	The interferometric scale factor is the number of waves per fringe.

	WavelengthIn as Single
	The wavelength, in meters, at which the interferogram was measured.

	ObliquityFactor as Single
	Phase correction factor required when using a Mirau objective on a microscope.  A value of 1.0 indicates no correction.

	NumAperture as Single
	Equals 1 / (2 * f‑number).

	AcquireType as Integer
	Indicates the setting of the Acquisition Mode control.  See ZygoAcquireType for values.

	IntensAvgCount as Integer
	The number of intensity averages performed.  Values of 0 or 1 indicate no averaging.

	PztCal as Integer
	Indicates whether or not the modulation amplitude was automatically adjusted during acquisition.  A value of 0 indicates that no adjustment was performed.

	PztGain as Integer
	Specifies the modulation amplitude value used during data acquisition.

	PartThickness as Single
	The thickness, in meters, of the part measured.

	AGC as Integer
	Indicates whether or not automatic gain control was performed during data acquisition.  A value of 0 indicates no AGC was performed.

	Sign as Integer
	Sign of the data.  0 is normal, 1 is inverted.

	Comment as String
	User‑entered remark line.  The value is truncated to 81 characters when the data file is saved.

	PartName as String
	User‑entered identified for the part measured.  The value is truncated to 39 characters when the data file is saved.

	PartSerNum as String
	User‑entered serial number for the part.  The value is truncated to 39 characters when the data file is saved.


Methods

	Copy(Source as MetroProDataAttr)
	Copies the values of Source.


MetroProDataFile (IMetroProDataFile interface)
Note: Multiple data files can be loaded by creating multiple MetroProDataFile objects and loading one data file with each object.

Methods

	LoadMetroProDataFile(


FileName as String,


pAttr as MetroProDataAttr

) as Integer
	Loads the data file specified by FileName.  The data files attributes are returned in pAttr.

A return value of 0 indicates no error.

	SaveMetroProDataFile(


FileName as String,


pAttr as MetroProDataAttr

) as Integer
	Saves the attributes specified by pAttr and the current phase and intensity data to the data file specified by FileName.

A return value of 0 indicates no error.

	ResetMetroProData()
	Removes phase and intensity data and resets all header and attribute values to their default values.

	GetMetroProPhaseData(


PhaseDataArray as Variant

) as Integer
	Returns the phase data in PhaseDataArray.  PhaseDataArray must either be a properly dimensioned 2D array of Longs or and empty Variant.

A return value of 0 indicates no error.

	SetMetroProPhaseData(


PhaseDataArray as Variant

) as Integer
	Sets the phase data to PhaseDataArray. 

A return value of 0 indicates no error.

	GetMetroProIntensData


IntensDataArray as Variant

) as Integer
	Returns the phase data in IntensDataArray. IntensDataArray must either be a properly dimensioned 3D array of Integers or and empty Variant.

A return value of 0 indicates no error.

	SetMetroProIntensData


IntensDataArray as Variant

) as Integer
	Sets the phase data to IntensDataArray. 

A return value of 0 indicates no error.


Installing the DLL

Manual Installation

Step 1:
Copy the DLL and the type library (TLB) onto you hard drive.  The files can be placed in any directory, but typical locations are the System32 directory in Windows directory or in the metropro\bin directory.

Step 2:
Register the DLL.  This can be done any number of ways.

· From the command line, type “regsvr32 <DLL_NAME>” where <DLL_NAME> is the full path and file name for the DLL.  If there are spaces in the path or filename, then the path and file name needs to be placed in double quotes (“).

· From Windows Explorer right‑click on the file and select “Open with…”.  If you are asked select a file to open the DLL, select regsvr32 or click on the “Other…” button and find regsvr32.  Finally click the “OK” button.

· From Visual Basic, select the “Project ( References…” menu item.  The click on the “Browse…” button.  Select the directory in which the type library is located (from step 1) and select the type library.  Finally click the “Open” button.

The DLL is now registered with your system.

Using the DLL

Visual Basic

Start a new project or open an existing project.  Select the “Project ( References…” menu item.  Check the “MetroPro DAT #.# Type Library” (where #.# is a version number) from the list box and click “OK”.

Microsoft Visual C++

Start a new project or open an existing project.  Open the “MFC Class Wizard” by pressing CTRL‑W or selecting the “View ( Class Wizard…” menu item.  Click the “Add Class…” button and select “From type library…”.  Select all the objects that you want available to your project and click OK.  Finally close the MFC Class Wizard.

The above procedure makes the objects available to the current project.

Microsoft Visual C++ 7
Start a new project or open an existing project.  Import classes from the type library by selecting the “Project ( Add Class…” menu item.  Select the “MFC Class From TypeLib” template and click Open.  Next, select the MetroProDAT type library from the list or browse to select the MetroProDAT.dll file.  Select all the objects that you want available to your project and click Finish.

The above procedure makes the objects available to the current project.

Accessing IZygoDataAttributes2 Attributes

The ZygoDataAttributes class supports both the IZygoDataAttributes interface and the IZygoDataAttributes2 interface.  By default, the IZygoDataAttributes interface is used, so to get access to the IZygoDataAttributes2 properties, the ZygoDataAttributes variable must be cast to an IZygoDataAttributes2 variable.

Visual Basic
Dim file As ZygoDataFile

Dim attr As ZygoDataAttributes

Dim attr2 As IZygoDataAttributes2

‘ Code to create and load the data file omitted

‘ Get the attributes using the IZygoDataAttributes interface

Set attr = file.Attribute

‘ Get the attributes using the IZygoDataAttributes2 interface

Set attr2 = attr
Microsoft Visual C++ 7
CZygoDataFile file;
// Code to create and load the data file omitted

// Get the attributes using the IZygoDataAttributes interface

CZygoDataAttributes attr(file.get_Attribute());

// Get the attributes using the IZygoDataAttributes2 interface

CZygoDataAttributes2 attr2(file.get_Attribute());

Reading A MetroPro Data File

The following sample code shows how to read a MetroPro data file and convert the phase data from zygos to waves.

Public Sub LoadData(ByVal FileName As String)

    Dim oData           As ZygoDataFile

    Dim Attr            As ZygoDataAttributes

    Dim Phase           As ZygoDataMatrix

    Dim Intens          As ZygoIntensityMatrix

    Dim ConvFactor      As Single

    Dim InvalidPhase    As Long

    Dim PhaseData()     As Long

    Dim PhaseF()        As Single

    Dim idx             As Long

    Dim jdx             As Long

    ' Create the data file object

    Set oData = New ZygoDataFile

    Set Attr = oData.Attribute

    ' Load the data

    oData.LoadData FileName

    ' Get the phase and intensity data

    Set Intens = oData.IntensityData

    Set Phase = oData.PhaseData

    ' Create the array of converted phase data

    ReDim PhaseF(LBound(Phase.Matrix, 1) To UBound(Phase.Matrix, 1), _

                 LBound(Phase.Matrix, 2) To UBound(Phase.Matrix, 2) _

                ) As Single

    ' Get the value of invalid data

    InvalidPhase = oData.InvalidPhase

    ' Get the conversion factor

    ConvFactor = Attr.IntfScaleFactor * Attr.ObliquityFactor

    If (Attr.PhaseRes = HighRes) Then

        ConvFactor = ConvFactor / 32768

    Else

        ConvFactor = ConvFactor / 4096

    End If

    ' Convert the phase data from zygos to waves

    PhaseData = Phase.Matrix

    For idx = LBound(PhaseF, 1) To UBound(PhaseF, 1)

        For jdx = LBound(PhaseF, 2) To UBound(PhaseF, 2)

            If (PhaseData(idx, jdx) >= InvalidPhase) Then

                PhaseF(idx, jdx) = 0

            Else

                PhaseF(idx, jdx) = ConvFactor * PhaseData(idx, jdx)

            End If

        Next jdx

    Next idx

    ' Delete the data file object

    Set oData = Nothing

End Sub

Backward Compatible Format

The following sample code shows how to read a MetroPro data file and convert the phase data from zygos to waves.  The bold line shows the only change needed to existing code.

Public Sub ReadMetroProDataFile(ByVal PathName As String)

    Dim Attr As MetroProDataAttr

    Dim nPhaseRows As Integer

    Dim nPhaseCols As Integer

    Dim nIntensRows As Integer

    Dim nIntensCols As Integer

    Dim nBuckets As Integer

    Dim Err As Integer

    Dim ErrMsg As String

    Dim IntensMat() As Long

    Dim PhaseMat() As Long

    Dim PhaseMatF() As Single

    Dim ZygosPerFringe As Long

    Dim PhaseVal As Long

    Dim ConvFactor As Single

    Dim nRow As Integer

    Dim nCol As Integer

    Dim InvalidInten As Long

    Dim InvalidPhase As Long

    Const NO_ERROR As Integer = 0

    Set Attr = New MetroProDataAttr

    InvalidInten = 64512

    InvalidPhase = 2147483640

    Err = NO_ERROR

    ' Load the data file

    Err = LoadMetroProDataFile(PathName, Attr)

    If (Err <> NO_ERROR) Then

        ErrMsg = "Error reading data file " & PathName

        MsgBox ErrMsg

        Exit Sub

    End If

    ' Set size of Intensity and Phase arrays based on

    ' row and column attributes returned in MetroProDataAttr parameter

    nPhaseRows = Attr.PhaseMatRows

    nPhaseCols = Attr.PhaseMatCols

    nIntensRows = Attr.IntensMatRows

    nIntensCols = Attr.IntensMatCols

    nBuckets = Attr.IntensMatBuckets

    If (nPhaseRows <> 0 And nPhaseCols <> 0) Then

        ' Resize phase matrices

        ReDim PhaseMat(0 To nPhaseCols - 1, 0 To nPhaseRows - 1)

        ReDim PhaseMatF(0 To nPhaseCols - 1, 0 To nPhaseRows - 1)

        ' Get the loaded phase matrix into PhaseMat

        Err = GetMetroProPhaseData(PhaseMat)

        If (Err <> NO_ERROR) Then Exit Sub

    End If

    If (nIntensRows <> 0 And nIntensCols <> 0) Then

        ' Resize intensity matrix

        ReDim IntensMat(0 To nIntensCols - 1, _

                        0 To nIntensRows - 1, _

                        0 To nBuckets - 1)

        ' Get the loaded intensity matrix into IntensMat

        Err = GetMetroProIntensData(IntensMat)

        If (Err <> NO_ERROR) Then Exit Sub

    End If

    ' Calculate factor to convert phase values

    ' from zygos to waves

    If Attr.PhaseRes = 0 Then

        ZygosPerFringe = 4096

    Else

        ZygosPerFringe = 32768

    End If

    ConvFactor = (Attr.IntfScaleFactor * Attr.ObliquityFactor) _

                 / ZygosPerFringe

    ' Convert phase values to waves

    For nCol = 0 To Attr.PhaseMatCols - 1

        For nRow = 0 To Attr.PhaseMatRows - 1

            PhaseVal = PhaseMat(nCol, nRow)

            If (PhaseVal < InvalidPhase) Then

                PhaseMatF(nCol, nRow) = ConvFactor * PhaseVal

            Else

                PhaseMatF(nCol, nRow) = 0!

            End If

        Next nRow

    Next nCol

End Sub

Writing A MetroPro Data File

The following sample code shows how to generate a MetroPro data file.  The phase data is generated as waves and is then converted to zygos.  The bold line shows the only change needed to existing code.

Public Sub SaveData(ByVal FileName As String)

    Dim oData       As ZygoDataFile

    Dim Attr        As ZygoDataAttributes

    Dim Phase       As ZygoDataMatrix

    Dim idx         As Long

    Dim jdx         As Long

    Dim PhaseData() As Long

    Dim PhaseF()    As Single

    Dim ConvFactor  As Single

    ' Create the data file object

    Set oData = New ZygoDataFile

    Set Attr = oData.Attribute

    ' Make changes to the attributes

    With Attr

        .CameraWidth = 320

        .CameraHeight = 240

        .LateralResolution = 0.000000001

        .PhaseRes = HighRes

        .RefractiveIndex = 1

        .PixelWidth = 0.000022

        .PixelHeight = 0.000022

    End With

    ' Resize the data matrix

    ReDim PhaseF(0 To Attr.CameraWidth - 1, _

                 0 To Attr.CameraHeight - 1) As Single

    ' Generate PhaseF matrix (in waves)

    ' The surface function is: Z = 0.001 * (col + row)

    ' which is a tilted plane

    For idx = LBound(PhaseF, 1) To UBound(PhaseF, 1)

        For jdx = LBound(PhaseF, 2) To UBound(PhaseF, 2)

            PhaseF(idx, jdx) = 0.001 * (idx + jdx)

        Next jdx

    Next idx

    ' Get the conversion factor

    ConvFactor = Attr.IntfScaleFactor * Attr.ObliquityFactor

    If (Attr.PhaseRes = HighRes) Then

        ConvFactor = 32768 / ConvFactor

    Else

        ConvFactor = 4096 / ConvFactor

    End If

    ' Resize the phase data

    ReDim PhaseData(0 To Attr.CameraWidth - 1, _

                    0 To Attr.CameraHeight - 1) As Long

    ' Convert the PhaseF data to zygos

    For idx = LBound(PhaseF, 1) To UBound(PhaseF, 1)

        For jdx = LBound(PhaseF, 2) To UBound(PhaseF, 2)

            PhaseData(idx, jdx) = ConvFactor * PhaseF(idx, jdx)

        Next jdx

    Next idx

    ' Save the phase data

    oData.PhaseData.Matrix = PhaseData

    ' Save the data file

    oData.SaveData FileName

    ' Delete the data file object

    Set oData = Nothing

End Sub

Backward Compatible Format

The following sample code shows how to generate a MetroPro data file.  The phase data is generated as waves and is then converted to zygos.  The bold line shows the only change needed to existing code.

Public Sub WriteMetroProDataFile(ByVal PathName As String)

    Dim Attr As MetroProDataAttr

    Dim PhaseMatF() As Single

    Dim PhaseMat() As Long

    Dim ZygosPerFringe As Long

    Dim PhaseVal As Long

    Dim ConvFactor As Single

    Dim C As Single

    Dim row As Integer

    Dim col As Integer

    Dim nRows As Integer

    Dim nCols As Integer

    Dim Err As Integer

    Const NO_ERROR As Integer = 0

    Set Attr = New MetroProDataAttr

    Err = NO_ERROR

    ' Clear any previous MetroPro data

    ' (matrices and attributes)

    ResetMetroProData

    ' Generate PhaseMatF matrix (in waves)

    ' The surface function is: Z = C * (col + row)

    ' which should be a tilted plane

    nCols = 320

    nRows = 240

    C = 0.001

    ReDim PhaseMat(0 To nCols - 1, 0 To nRows - 1) As Long

    ReDim PhaseMatF(0 To nCols - 1, 0 To nRows - 1) As Single

    For col = 0 To nCols - 1

        For row = 0 To nRows - 1

            PhaseMatF(col, row) = C * (col + row)

        Next row

    Next col

    ' Assign required attribute values

    Attr.CameraCols = nCols

    Attr.CameraRows = nRows

    Attr.CameraRes = 0!

    Attr.PhaseRes = 1

    Attr.IntfScaleFactor = 0.5

    Attr.ObliquityFactor = 1!

    ' Calculate factor to convert phase values

    ' from waves to zygos

    If (Attr.PhaseRes = 0) Then

        ZygosPerFringe = 4096

    Else

        ZygosPerFringe = 32768

    End If

    ConvFactor = ZygosPerFringe / _

                (Attr.IntfScaleFactor * Attr.ObliquityFactor)

    ' Convert phase values to zygos and load PhaseMat matrix

    For col = 0 To nCols - 1

        For row = 0 To nRows - 1

            PhaseMat(col, row) = PhaseMatF(col, row) * ConvFactor

        Next row

    Next col

    ' Set PhaseMat as the matrix to be saved in the file

    Err = SetMetroProPhaseData(PhaseMat)

    ' Save the data file

    Err = SaveMetroProDataFile(PathName, Attr)

    If (Err <> NO_ERROR) Then

        MsgBox "Unable to save MetroPro Data file.", vbOKOnly, "Error"

    End If

End Sub

